version 18.0               // version control
set processors 8           // to ensure replicability across different numbers of cores
clear all                  // clear existing data
macro drop _all            // and macros, clean slate
set seed 20220909          // set seed
/* 
Notes: 
1) Must put "set sortseed #" right on top of EACH "telasso" (rather than at the top of the do file) to ensure the same number of selected controls will be reported.
2) For detailed explanations, see Stata's technical support's email chain on 10/10/2023. 
*/

local pgm  "dst-Appendix_Table_A7_Panel_B_mental_distress_telasso_check_three_variables" // file name
local who  "Muzhe Yang"                                                                  // author
local dte  "2025-01-22"                                                                  // created date
local dte2 "`c(current_date)'"                                                           // last run date
local tag  "`pgm'.do, created by `who' on `dte', last run on `dte2'"

capture log close
log using "code\analysis\tables\appendix\\`pgm'.txt", text replace 
display "`tag'"

**# data prep ------------------------------------------------------------------------------------

use "data_clean\dst-data04_for_estimation_within_250_miles", clear

*## (1) create the 9 cells, following page 215 of the following paper:
*## Giuntella, O. and F. Mazzonna (2019). "Sunset Time and the Economic Effects of Social Jetlag: Evidence from US Time Zone Borders." Journal of Health Economics 65: 210-226.

assert !missing(centroid_lat)
assert !missing(region)
gen     cell = 1 if centroid_lat > 40                        & region == "eastern and central"
replace cell = 2 if (34 < centroid_lat & centroid_lat <= 40) & region == "eastern and central"
replace cell = 3 if centroid_lat <= 34                       & region == "eastern and central"
replace cell = 4 if centroid_lat > 40                        & region == "central and mountain"
replace cell = 5 if (34 < centroid_lat & centroid_lat <= 40) & region == "central and mountain"
replace cell = 6 if centroid_lat <= 34                       & region == "central and mountain"
replace cell = 7 if centroid_lat > 40                        & region == "mountain and pacific"
replace cell = 8 if (34 < centroid_lat & centroid_lat <= 40) & region == "mountain and pacific"
replace cell = 9 if centroid_lat <= 34                       & region == "mountain and pacific"
tab cell time_zone, missing
summ centroid_lat if cell == 1 | cell == 4 | cell == 7
summ centroid_lat if cell == 2 | cell == 5 | cell == 8
summ centroid_lat if cell == 3 | cell == 6 | cell == 9

*## (2) control variables

global x "pop white_pct black_pct hispanic_pct pop_under_18yr_pct pop_65yr_over_pct age_median educ_hs_pct educ_coll_pct married_pct hh_size hh_income_median home_value_median no_health_ins_pct unemploy_pct"
global cvars    c.($x dist_to_border centroid_lat daylight_dur_max)##c.($x dist_to_border centroid_lat daylight_dur_max)
global controls $cvars i.cell i.cell#c.($x dist_to_border centroid_lat daylight_dur_max) i.wave i.wave#c.($x dist_to_border centroid_lat daylight_dur_max)

global cvars_drop_daylight_dur_max    c.($x dist_to_border centroid_lat)##c.($x dist_to_border centroid_lat)
global controls_drop_daylight_dur_max $cvars_drop_daylight_dur_max i.cell i.cell#c.($x dist_to_border centroid_lat) i.wave i.wave#c.($x dist_to_border centroid_lat)

global cvars_drop_centroid_lat    c.($x dist_to_border daylight_dur_max)##c.($x dist_to_border daylight_dur_max)
global controls_drop_centroid_lat $cvars_drop_centroid_lat i.cell i.cell#c.($x dist_to_border daylight_dur_max) i.wave i.wave#c.($x dist_to_border daylight_dur_max)

global controls_drop_cell_dummies $cvars i.wave i.wave#c.($x dist_to_border centroid_lat daylight_dur_max)

global cvars_only_daylight_dur_max c.($x dist_to_border daylight_dur_max)##c.($x dist_to_border daylight_dur_max)
global controls_only_daylight_dur_max $cvars_only_daylight_dur_max i.wave i.wave#c.($x dist_to_border daylight_dur_max)

global cvars_only_centroid_lat c.($x dist_to_border centroid_lat)##c.($x dist_to_border centroid_lat)
global controls_only_centroid_lat $cvars_only_centroid_lat i.wave i.wave#c.($x dist_to_border centroid_lat)

global controls_only_cell_dummies c.($x dist_to_border)##c.($x dist_to_border) i.cell i.cell#c.($x dist_to_border) i.wave i.wave#c.($x dist_to_border)

**# estimation prep -------------------------------------------------

encode CountyFIPS, gen(county_fips)
egen nomiss = rowmiss($x dist_to_border centroid_lat daylight_dur_max)
assert !missing(StateAbbr)
local radius = 50
gen sample_selection = (nomiss == 0 & StateAbbr != "AZ" & dist_to_border <= `radius')
global xfolds_resample "xfolds(10) resample(3) nolog"

**# table ---------------------------------------------------------------------------------

set sortseed 12102023
telasso (mental_hlth_distress $controls_only_centroid_lat) (treat $controls_only_centroid_lat) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Included"
scalar dayh_sel = "Excluded"
scalar cell_sel = "Excluded"
quietly etable, replace ///
        column(index) ///
        keep(r1vs0.treat) ///
        cstat(_r_b,  nformat(%9.3f)) ///
        cstat(_r_se, nformat(%9.3f) label("(Standard error)")) ///
        cstat(_r_p,  nformat(%9.3f) label("[P-value]")) ///
        mstat(N, nformat(%9.0fc) label("Number of observations")) ///
        mstat(case_lati_sel = lati_sel, label("Latitude of a census tract's centroid")) ///
        mstat(case_dayh_sel = dayh_sel, label("Maximum of daylight hours per day, county level")) ///
        mstat(case_cell_sel = cell_sel, label("Cell dummy variables")) ///
        stars(0.10 "*" 0.05 "**" 0.01 "***", attach(_r_b) decreasing pvname("p-value") nformat(%9.2f)) showstars showstarsnote ///
        novarlabel nocenter 

set sortseed 12102023
telasso (mental_hlth_distress $controls_only_daylight_dur_max) (treat $controls_only_daylight_dur_max) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Excluded"
scalar dayh_sel = "Included"
scalar cell_sel = "Excluded"
quietly etable, append

set sortseed 12102023
telasso (mental_hlth_distress $controls_only_cell_dummies) (treat $controls_only_cell_dummies) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Excluded"
scalar dayh_sel = "Excluded"
scalar cell_sel = "Included"
quietly etable, append

set sortseed 12102023
telasso (mental_hlth_distress $controls_drop_centroid_lat) (treat $controls_drop_centroid_lat) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Excluded"
scalar dayh_sel = "Included"
scalar cell_sel = "Included"
quietly etable, append

set sortseed 12102023
telasso (mental_hlth_distress $controls_drop_daylight_dur_max) (treat $controls_drop_daylight_dur_max) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Included"
scalar dayh_sel = "Excluded"
scalar cell_sel = "Included"
quietly etable, append

set sortseed 12102023
telasso (mental_hlth_distress $controls_drop_cell_dummies) (treat $controls_drop_cell_dummies) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Included"
scalar dayh_sel = "Included"
scalar cell_sel = "Excluded"
quietly etable, append

set sortseed 12102023
telasso (mental_hlth_distress $controls) (treat $controls) if sample_selection == 1, vce(cluster county_fips) $xfolds_resample rseed(10101)
scalar lati_sel = "Included"
scalar dayh_sel = "Included"
scalar cell_sel = "Included"
quietly etable, append

collect title "Table: The Impact of Circadian Misalignment Estimated by the Double/Debiased ML, Checking Three Variables" 

collect layout 
collect addtags top_row[1]
collect layout (colname[r1vs0.treat]#result[_r_b _r_se _r_p] result[N case_lati_sel case_dayh_sel case_cell_sel]) ///
               (top_row[1]#cmdset#stars)

collect label levels top_row ///
        1 "Treat = 1 for census tracts located east of the time zone border; Treat = 0 for census tracts located west of the time zone border", modify
collect label levels cmdset 1 "(1)" 2 "(2)" 3 "(3)" 4 "(4)" 5 "(5)" 6 "(6)" 7 "(7)", modify
collect label levels colname r1vs0.treat "Treat (1/0)", modify 

collect style header top_row, level(label)
collect style header result, level(label)
collect style header result[_r_se], level(label)
collect style header result[_r_p], level(label)
collect style header cmdset, level(label)
collect style header colname, level(label)
collect style cell result[_r_b N], sformat(%s)
collect style cell result[_r_se], sformat("(%s)")
collect style cell result[_r_p], sformat("[%s]")
collect style cell border_block[corner], border(top, pattern(double))
collect style cell border_block[column-header], border(top, pattern(double))
collect style cell border_block[row-header], border(bottom, pattern(double))
collect style cell border_block[item], border(bottom, pattern(double))

collect export "code\analysis\tables\appendix\\`pgm'.xlsx", replace

log close
exit